RANK

PURPOSE

Compute the ranks of a variable.

DESCRIPTION

Ranks for a variable containing N elements are determined by first sorting the data from lowest to highest. Then the element with the smallest value is assigned a rank of 1, the element with the second smallest value is assigned a rank of 2, and so on until the maximum value is assigned a rank of N.

SYNTAX

LET <var> = RANK <y1> <SUBSET/EXCEPT/FOR qualification>
where <y1> is a response variable;
<var> is a variable where the computed ranks are saved;
and where the <SUBSET/EXCEPT/FOR qualification> is optional.

EXAMPLES

LET A = RANK Y1

NOTE 1

Ties are assigned an average rank. For example, if the 2nd and 3rd highest values are equal, each is assigned a rank of 2.5.

NOTE 2

Many nonparametric statistical methods are based on ranks. The program sample shows the use of ranks in computing the Kruskal-Wallis test for one-way ANOVA.

DEFAULT

None

SYNONYMS

None

RELATED COMMANDS

SORT	=	Sort the elements in a variable.
SORTC	=	Sort the elements in a variable and carry one or more variables along.

APPLICATIONS

Nonparametric statistics

IMPLEMENTATION DATE

Pre-1987

PROGRAM

```
. Perform a Kruskal-Wallis non-parametric 1-way ANOVA
. Data from “Probability and Statistics for Engineers and
. Scientists” by Walpole and Myers.
LET P = 3
LET X1 = DATA 24.0 16.7 22.8 19.8 18.9
LET X2 = DATA 23.2 19.8 18.1 17.6 20.2 17.8
LET X3 = DATA 18.4 19.1 17.3 17.3 19.7 18.9 18.8 19.3
.
LOOP FOR K = 1 1 P
      LET N^K = SIZE X^K
      LET TAG^K = K FOR I = 1 1 N^K
END OF LOOP
LOOP FOR L = 2 1 P
      LET TEMP = X^L
      EXTEND X1 TEMP
      LET TEMP = TAG^L
      EXTEND TAG1 TEMP
      DELETE TEMP
END OF LOOP
.
LET N = SIZE X1
LET R = RANK X1
LET SUM1 = 0
LOOP FOR K = 1 1 P
      LET R^K = SUM R SUBSET TAG1 = K
      LET SUM1 = SUM1 + (R^K)**2/N^K
END OF LOOP
.
LET H = SUM1*(12/(N*(N+1))) - 3*(N+1)
LET ALPHA = 0.95
LET DF = P - 1
LET CRITICAL = CHSPPF(ALPHA,DF)
.
PRINT “ “
PRINT “H0: ^P INDEPENDENT SAMPLES ARE FROM IDENTICAL POPULATIONS”
PRINT “HA: ^P INDEPENDNET SAMPLES ARE FROM DIFFERENT POPULATIONS”
PRINT “KRUSKAL-WALLIS H STATISTIC = ^H”
PRINT “CHI-SQUARE CRITICAL VALUE = ^CRITICAL”
IF H <= CRITICAL
      PRINT “ACCEPT NULL HYPOTHESIS”
END OF IF
IF H > CRITICAL
 PRINT “REJECT NULL HYPOTHESIS”
END OF IF
```

The following output is generated.

```
H0: 3 INDEPENDENT SAMPLES ARE FROM IDENTICAL POPULATIONS
HA: 3 INDEPENDNET SAMPLES ARE FROM DIFFERENT POPULATIONS
KRUSKAL-WALLIS H STATISTIC = 1.658619
CHI-SQUARE CRITICAL VALUE = 5.991465
ACCEPT NULL HYPOTHESIS
```